Charts


Go to Page 5.
· Make the following chart. From the Ribbon choose Insert, Columns,2DColumn


 







(stats.xlsm: charts sheet)

We have produced an embedded chart.

To Produce an Embedded Chart using VBA
· Place a button on the sheet and write this: ie Using a Chart object:
Private Sub cmdChart_Click()

Dim cht As Chart
Set cht = Charts.Add

Set cht = cht.Location(Where:=xlLocationAsObject, Name:="charts")

End Sub

The above code will produce an empty chart container
 as shown here since we  have not specified the data yet.

Equally we may produce a chart using the Shapes object as follows:

Private Sub cmdChart_Click()

ie Using a Shape object:
Dim shp As Shape
Set shp = ActiveSheet.Shapes.AddChart

End Sub

 This is the method used by the macro recorder.
 Presumably it is easier for the recorder to cope
 with shapes generically? so we indeed will from now on 

 use this method to create any new charts. 
If we already have a chart on our worksheet (eg  if we have run the code above) we can then refer to it as follows:

Recall that there is also a container of a chart.
In VBA this is a ChartObject.

Private Sub cmdChart_Click()
Dim cht As Chart, cho As ChartObject

Set cho = ActiveSheet.ChartObjects(1)

'Set cht = cho.Chart

cho.Select   '(cht.Select won’t work.)
End Sub
We could also refer to the ChartObjects by name eg chart1 or whatever its name is.

Set cho = ActiveSheet.ChartObjects("chart1").
Solution to previous exercise: MsgBox cho.Name
Selecting using the Mouse: 

First make a chart using this data in Excel.

	
	mon
	tue
	wed
	thu
	fri

	week1
	32
	45
	23
	34
	22

	week2
	45
	33
	11
	22
	44

	week3
	33
	45
	65
	32
	12






Private Sub cmdChart1_Click()

'Using Charts Collection
Dim cht As Chart

Set cht = Charts.Add
Set cht = cht.Location(Where:=xlLocationAsObject, Name:="charts")

‘Must set the location else it will be a chart sheet.

‘ ie use Shapes!

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4")

End Sub
Private Sub cmdCharts2_Click()

'Using Shapes Collection
Dim shp As Shape, cht As Chart

Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4")

End Sub

To Refer to the Series of a chart already on the worksheet.
Private Sub cmdChart_Click()

Dim cht As Chart, cho As ChartObject

Dim src As SeriesCollection, i As Integer
Set cho = ActiveSheet.ChartObjects(1)
Set cht = cho.Chart

Set src = cht.SeriesCollection

For i = 1 To src.Count

    MsgBox src(i).Name
Next i

End Sub
result:






    week2 etc.

Note that in Excel the corresponding series string will automatically appear in the formula bar when the series is selected:


This string is composed of 3 bits:



=SERIES(charts!$A$2,charts!$B$1:$F$1,charts!$B$2:$F$2,1)



Reading the values of these series is not as easy as it looks. This string has to be parsed!
(See the ChartSeries Inf Class.xls by Walchenbach - see the SERIESFormulaElement function on the CharSeries class module.)

Charts: 

To Delete any Existing Charts
 On Error Resume Next
 ActiveSheet.ChartObjects.Delete

 On Error GoTo 0

To Assign a Data Series (Using a Range)
We will produce a new chart using code and then assign a data series:

· First delete any existing charts manually (or include the above code) and run this code:
Private Sub cmdChart_Click()

Dim shp As Shape, cht As Chart

Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4") 

End Sub






Note that in doing the above, Excel has cleverly guessed at:
1. The X Values (ie the CATEGORIES) to be the respective headings of the columns of the s/sheet ie mon, tue, wed, thu, fri.

3. The Y Values for each category to be the respective columns of data in the s/sheet. SERIES.
1. The Names of the series to be week1, week2 & week3  ie the headings of the rows of the s/sheet..
eg if we highlight the first row of data in the chart above (by clicking on just one of them) we get the following series string in the formula bar:

=SERIES(charts!$A$2,charts!$B$1:$F$1,charts!$B$2:$F$2,1)
recall that the format is:

=SERIES([Series Name],[X Values],[Y Values],[Plot Order])

· If the data is arranged in columns change the line in the previous code to:

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4"), PlotBy:=xlColumns

You should see the axes of the graph switch as well as the series names change as follows:








(recall the chart:)

ie the SERIES are now the respective Columns.

The default was xlRows where the series were the respective rows as previously mentioned  .
Recall that swapping rows and columns is equivalently achieved in Excel.

by Right-clicking and Select Data and then Switch Row/Column.



To provide  a Data Series in code using an Array
As previously mentioned when adding a new chart we will use the Shape object.

(This will emulate the chart we added from Excel at the very start of this manual.)

Private Sub cmdChart_Click()

Dim shp As Shape, cht As Chart, sr As Series


Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "week1" 

sr.XValues = Array("mon", "tue", "wed", "thu", "fri")

sr.Values = Array(32, 45, 23, 34, 22)

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "week2" 

sr.Values = Array(45, 33, 11, 22, 44)

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "week3" 

sr.Values = Array(33, 45, 65, 32, 12)

End Sub

Note that the series string (see the formula bar) now relates to the array rather than a range,
=SERIES("week2",,{45,33,11,22,44},2)
To Reverse the Series Order 
(We can’t simply use something like:

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4"), PlotBy:=xlColumns 
since we are not using the above technique to set the Data Source.)
Using the Shape object:
Private Sub cmdChart_Click()

Dim shp As Shape, cht As Chart, sr As Series

Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "Mon" 'off1
sr.XValues = Array("week1", "week1", "week1")

sr.Values = Array(32, 45, 33)

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "Tue" 'off1
sr.Values = Array(45, 33, 45)

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "wed" 'off1
sr.Values = Array(23, 11, 65)


Set sr = cht.SeriesCollection.NewSeries

sr.Name = "thu" 'off1
sr.Values = Array(34, 22, 32)

Set sr = cht.SeriesCollection.NewSeries

sr.Name = "fri"

sr.Values = Array(22, 44, 12) 

End Sub
 
Note the series string:   =SERIES("Tue",,{45,33,45},2)
We could use a Variant array:
Dim arr As Variant
arr = Range("C2:c4")

sr.Values = arr

To Position the Chart
Note that the Shapes object (or the ChartObject can also be defined and used) as below to position the chart (rather than the Chart object) since it is (also) the container.

Dim shp As Shape, rng As Range


Set shp = ActiveSheet.Shapes.AddChart

Set rng = Range("g6:k12")

'Position Shape over range
shp.Top = rng.Top

shp.Left = rng.Left

shp.Height = rng.Height

shp.Width = rng.Width


Note the size of the chart container is also determined by the range:  ("g6:k12") in this case.
To Name Our Chart Object. Also a property of the "container":
shp.Name = "MangoesChart"
To Define The Chart Type
Private Sub cmdChart_Click()

Dim shp As Shape, cht As Chart

Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4")

cht.ChartType = xlColumnStacked

End Sub

Note the itellisense help: 

To Add a Title 
A property of the Chart object (and not the ChartObject object).
cht.HasTitle = True
cht.ChartTitle.Text = "Month1"
Data Labels
Not using code:

To add data labels we can of course right-click on the chart…







Data Labels continued.
Using VBA:

Private Sub cmdDataLabels_Click()

Dim shp As Shape, cht As Chart, sr As Series
Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4")

cht.ChartType = xlColumnStacked

Set sr = cht.SeriesCollection(1)

sr.HasDataLabels = True
End Sub



Points
Each SeriesCollection has a Points collection – eg in the previous chart, SeriesCollection(1) has elements  32, 45,23,34,22.
This code will iterate through the Points collection of SeriesCollection(1) of the chart above.
Private Sub cmdDataLabels_Click()

Dim shp As Shape, cht As Chart, sr As Series, pt As Point, i As Integer
Set shp = ActiveSheet.Shapes.AddChart

Set cht = shp.Chart

cht.SetSourceData Source:=ActiveSheet.Range("A1:f4")

cht.ChartType = xlColumnStacked

Set sr = cht.SeriesCollection(1)

sr.HasDataLabels = True

For i = 1 To sr.Points.Count 


  (sr.Points.Count=5 in this case.)
    Set pt = sr.Points(i)

    MsgBox pt.DataLabel.Text


     …45,23,34,22.
Next i

End Sub
Keep in mind that the Points collection only gives us access to the "cosmetic" properties ie labels of the chart series. In particular access to the DataLabel’s - as above. We could for example set the text of an individual data label using the .Text property (or hide it using .HasDataLabel = False )
To confirm this  that take a look at the properties and methods using Itellisense:



As mentioned, if we were to change the Point.Datalabel value it would only change the value on the chart- it would not change the underlying value. (To do that we would use the 

MsgBox sr.Values(1) 

· Try it. )
Beware don’t try: 
sr.Values(1) = 10 ! It seems that we can’t poke the Values property directly without using an array? This causes a serious crash.

The following example produces an array of values in code which are then applied to a chart.

Private Sub cmdAddSeries_Click()

Dim cbo As ChartObject, cht As Chart, sc As SeriesCollection

Dim sr As Series, i As Integer, arr As Variant, c As Integer
Dim vals

Dim serArray() As Double
'arr = Range("c3:c5").Value
'use a variant and simply extend it by reassigning it?
AddJustChart "F22:K30"

c = ActiveSheet.ChartObjects.Count

Set cbo = ChartObjects(c)

Set cht = cbo.Chart

Set sc = cht.SeriesCollection

'Delete all existing chart series
For i = sc.Count To 1 Step -1

    sc(i).Delete

Next i

cht.ChartType = xlColumnStacked

Set sr = cht.SeriesCollection.NewSeries 'Repeat this for as many OFF periods
ReDim serArray(3)

serArray(1) = 10 'these are X Values - not Series
serArray(2) = 15 'Base values so to speak.
serArray(3) = 20

sr.Name = "off1"

sr.Values = serArray

sr.HasDataLabels = True
Set sr = cht.SeriesCollection.NewSeries

'ReDim serArray(3)
serArray(1) = 20  'These are new values 'on top of' the previous values.
serArray(2) = 25 ' ie for the NEXT DAY
serArray(3) = 10 'so we'll have to iterate thru all the days first !!
sr.Name = "off2" '  and pick up the first elements and then all the second elements etc
sr.Values = serArray ' (provided there ARE second elements) etc
sr.HasDataLabels = True
Set sr = cht.SeriesCollection.NewSeries

'ReDim serArray(3)
serArray(1) = 10  'X Values - not Series
serArray(2) = 0 'Base values so to speak.
serArray(3) = 20

sr.Name = "off3"

sr.Values = serArray

sr.HasDataLabels = True
'add one more series
Set sr = cht.SeriesCollection.NewSeries

'ReDim serArray(4)
serArray(1) = 0  'X Values - not Series
serArray(2) = 0 'Base values so to speak.
serArray(3) = 15

sr.Name = "off4"

sr.Values = serArray

sr.XValues = Array("01/01/2013", "02/01/2013", "03/01/2013")

'of course we would take these from the w/sheet.
sr.HasDataLabels = True
'todo : find the ON periods and plot those too.
End Sub
Sub AddJustChart(st As String)

Dim shp As Shape

    'Delete any existing ChartObjects
'    On Error Resume Next
'    ActiveSheet.ChartObjects.Delete
'    On Error GoTo 0
    'Create new embedded chart
    Set shp = ActiveSheet.Shapes.AddChart

    'Position Shape over range
    With Range(st)

        shp.top = .top

        shp.Left = .Left

        shp.Height = .Height

        shp.Width = .Width

    End With
End Sub
  For interesting examples of dynamic series see:

Private Sub cmdAddDataLabels_Click()

Dim ch As Chart, chObj As ChartObject, src As Object, r As Integer
Dim pt As Point, rngGammaValues As Range

Set chObj = ActiveSheet.ChartObjects(1)

Set ch = chObj.Chart

Set src = ch.SeriesCollection(1)

src.ApplyDataLabels      ' to create datalabels just in case
Set rngGammaValues = Range("c15:c34")
For r = 1 To rngGammaValues.Count

    src.Points(r).DataLabel.Text = "g = " & rngGammaValues.Cells(r).Value

Next r

End Sub

See stats.xlsm labels sheet
see also dynamic labels (on dynamic labels sheet stats.xslm):
Private Sub Worksheet_Change(ByVal Target As Range)

Dim r As Integer
If Target.Address = "$I$18" Then
    With ActiveSheet.ChartObjects(1).Chart

        For r = 1 To Range("Table1[Country]").Rows.Count

            If Range("Table1[Region]").Cells(r).Value = Range("$I$18") Then
                .SeriesCollection(1).Points(r).Format.Fill.ForeColor.RGB = RGB(0, 0, 255)

                .SeriesCollection(1).Points(r).ApplyDataLabels

                .SeriesCollection(1).Points(r).DataLabel.Text = Range("Table1[Country]").Cells(r).Value

            Else
                .SeriesCollection(1).Points(r).Format.Fill.ForeColor.RGB = RGB(198, 198, 198)

                On Error Resume Next
                .SeriesCollection(1).Points(r).DataLabel.Delete

                On Error GoTo 0

            End If
        Next r

    End With
End If
End Sub

Extra notes:
.ValuesType
We can use extend to extend a data series from the worksheet:

Use the SeriesCollection method to return the SeriesCollection collection. The following example adds the data in cells C1:C10 on worksheet one to an existing series in the series collection in embedded chart one.

Worksheets(1).ChartObjects(1).Chart. _

    SeriesCollection.Extend Worksheets(1).Range("c1:c10")
ref : Chapter 8 : Excel2007ProgReference.pdf in Manual folder)

also: ChartSeriesInfo Class.xls
see Chart2.xlsm: By John Green etc:

To Place a Chart on a UserForm
We must export our chart as an image and then load the image into an Image frame on the userform.


Private Sub Cmdloadimage_Click()

Dim Mychart As Chart

Dim Chartdata As Range

Dim Chartindex As Integer
Dim Chartname As String
Dim Imgname As String
Set Mychart = Activesheet.Shapes.Addchart(Xlxyscatterlines).Chart

Mychart.Seriescollection.Newseries

'Mychart.Seriescollection(1).Name = "Eds"

Mychart.Seriescollection(1).Values = Range("Rngdataref").Offset(, 1)

Mychart.Seriescollection(1).Xvalues = Range("Rngdataref")

Imgname = Application.Defaultfilepath & Application.Pathseparator & "Tempchart.Gif"

Mychart.Export Filename:=Imgname

Frmselect.Image1.Picture = Loadpicture(Imgname)

Activesheet.Chartobjects(1).Delete

End Sub
see 

http://www.homeandlearn.org/create_vba_chart_image.html
Note the Name of this chart.





The Series Names are taken from the row headings.  They are used for the legend.











eg series 1 (Blue)


(Named week1)








Each category (mon. tue, …) ie columns in the original s/sheet is a “cluster of individual series” ie rows in the original s/sheet.





Click on the edge to see that we can select the container of the chart (as shown below) or click on the chart itself to show that can be selected as well - inside the container. 


�


Note that the Ribbon menu changes to show a Chart Tools tag.





Excel assumes that the Series will be in rows (ie horizontal) and the individual categories in the columns. 


To change this behavior right-click on the chart and choose Select Data and then click  the Switch Rows/ Column button.





Chart sheet:


We may also place a chart on separate Chart Sheet as follows: Select the chart and from the menu choose the Design tab and then Move Chart Location. Choose a new sheet:





�









































Only a (separate) chartsheet as shown here can have events:


ie the chart as produced above cannot (Try double-clicking on the above chart!) - and to call these events we need to place an old-fashioned Forms button on the chart at that!


�








(If we omit the location then a new chart sheet will be created instead! - by default)





Ex: Find the name of the chart.





Tip: To make a reference to the ChartObjects from the Chart object we could use the Parent property eg try 


MsgBox cht.Parent.





Ctrl-click to select the ChartObject object.





Click to select the Chart object.





Use the chart we just constructed.





All of the series.





The Names of the series.





The series name. (The values here are also used for the legend.)





The x series.





The series number.





The y series.





(Start lesson here:)





(Default for Series is rows.)





Rule: Rows –X-axis.





The default was xlRows





It is actually best to clear out all of these initial series, and start from scratch with the series you intend to add. This macro clears the series of the chart:


Sub RemoveUnwantedSeries()


    With ActiveChart


        Do Until .SeriesCollection.Count = 0


            .SeriesCollection(1).Delete


        Loop


    End With


End Sub





Series in row





Series in column.





cht.SetDataSource ?





The XValues ie the values along the  X-Axis could be anything - not related to the data!





The respective series could also be arbitrarily named using .Name -unrelated to the data!





But the individual series are absolutely defined by the Values array/range - selected as either rows of columns.





Series are plotted "respectively" for each X-Axis group.





… and choose Add Data Labels.





To modify the label's formatting etc right-click on a label and choose Format Data Series and then use the dialog box below to modify..





Series 1 has data labels.


 To give all the three series labels we could loop through the SeriesCollection.





Type a full-stop and then take a look.

